```
> restart:
```

```
> with(inttrans):
```

Sistema di tre equazioni differenziali del secondo ordine

```
> eq1 := (D@@2)(x)(t)+(D@@2)(y)(t)+(D@@2)(z)(t)=1;
```

$$eq1 := D^{(2)}(x)(t) + D^{(2)}(y)(t) + D^{(2)}(z)(t) = 1$$

```
> eq2 := (D@@2)(x)(t)-3*(D@@2)(y)(t)-5*(D@@2)(z)(t)=exp(t);
```

$$eq2 := D^{(2)}(x)(t) - 3\,D^{(2)}(y)(t) - 5\,D^{(2)}(z)(t) = \mathbf{e}^{t}$$

```
> eq3 :=
  (D@@2)(x)(t)-D(x)(t)-(D@@2)(y)(t)-(D@@2)(z)(t)+3*D(z)(t)=exp(t)*sin(3*t);
```

$$eq3 := D^{(2)}(x)(t) - D(x)(t) - D^{(2)}(y)(t) - D^{(2)}(z)(t) + 3\,D(z)(t) = \mathbf{e}^{t}\,\sin(3\,t)$$

Condizioni Iniziali

```
> ini := [x(0)=1,y(0)=-1,z(0)=1,D(x)(0)=0,D(y)(0)=-10,D(z)(0)=1];
```

$$ini := [x(0) = 1, y(0) = -1, z(0) = 1, D(x)(0) = 0, D(y)(0) = -10, D(z)(0) = 1]$$

Cambio di variabile per migliorare la leggibilità

```
> SUBS :=
  {laplace(x(t),t,s)=x(s),laplace(y(t),t,s)=y(s),laplace(z(t),t,s)=z(s)};
```

$$SUBS := \{laplace(y(t), t, s) = y(s), laplace(z(t), t, s) = z(s), laplace(x(t), t, s) = x(s)\}$$

Sistema Trasformato

```
> eq1T := subs(SUBS,laplace(eq1,t,s));
  eq2T := subs(SUBS,laplace(eq2,t,s));
  eq3T := subs(SUBS,laplace(eq3,t,s));
```

$$eq1T := s^2\,x(s) - D(x)(0) - s\,x(0) + s^2\,y(s) - D(y)(0) - s\,y(0) + s^2\,z(s) - D(z)(0) - s\,z(0) = \frac{1}{s}$$

$$eq2T := s^2\,x(s) - D(x)(0) - s\,x(0) - 3\,s^2\,y(s) + 3\,D(y)(0) + 3\,s\,y(0) - 5\,s^2\,z(s) + 5\,D(z)(0) + 5\,s\,z(0) = \frac{1}{s-1}$$

$$eq3T := s^2\,x(s) - D(x)(0) - s\,x(0) - s\,x(s) + x(0) - s^2\,y(s) + D(y)(0) + s\,y(0) - s^2\,z(s) + D(z)(0) + s\,z(0) + 3\,s\,z(s) - 3\,z(0) = \frac{3}{s^2 - 2\,s + 10}$$

Sostituzione delle condizioni al contorno

```
> eq1TT := subs(ini,eq1T);
  eq2TT := subs(ini,eq2T);
  eq3TT := subs(ini,eq3T);
```

$$eq1TT := s^2\,x(s) + 9 - s + s^2\,y(s) + s^2\,z(s) = \frac{1}{s}$$

$$eq2TT := s^2\,x(s) - 25 + s - 3\,s^2\,y(s) - 5\,s^2\,z(s) = \frac{1}{s-1}$$

$$eq3TT := s^2\,x(s) - 11 - s - s\,x(s) - s^2\,y(s) - s^2\,z(s) + 3\,s\,z(s) = \frac{3}{s^2 - 2\,s + 10}$$

Ho ottenuto un sistema algebrico equivalente.

Risoluzione del sistema lineare rispetto a x(s) e y(s)

```
> SOL := solve({eq1TT,eq2TT,eq3TT},{x(s),y(s),z(s)});
```

$$SOL := \left\{ x(s) = \frac{14\,s^4 + 110\,s^3 + 4\,s^6 - 2\,s^5 - 187\,s^2 + 178\,s - 90}{2\,s^3\,(2\,s^4 - s^3 + 9\,s^2 + 40\,s - 50)}, \right.$$

$$y(s) = -\frac{-13\,s^4 + 177\,s^3 + 2\,s^6 + 192\,s^2 - 358\,s - 10 + 19\,s^5}{s^3\,(2\,s^4 - s^3 + 9\,s^2 + 40\,s - 50)},$$

$$\left. z(s) = \frac{4\,s^6 + 2\,s^5 + 160\,s^3 - 231\,s^2 + 86\,s - 30}{2\,s^3\,(2\,s^4 - s^3 + 9\,s^2 + 40\,s - 50)} \right\}$$

Sostituzione per migliorare la leggibilità

```
> SUBSINV := {invlaplace(x(s),s,t)=x(t),
              invlaplace(y(s),s,t)=y(t),
              invlaplace(z(s),s,t)=z(t)};
```

$$SUBSINV := \{invlaplace(x(s), s, t) = x(t), invlaplace(y(s), s, t) = y(t), invlaplace(z(s), s, t) = z(t)\}$$

```
> subs(SUBSINV,invlaplace(SOL,s,t)) ;
```

$$\left\{ y(t) = -\frac{513}{250} + \frac{16356}{14875}\,\mathbf{e}^{\left(-\frac{5}{2}t\right)} - \frac{183}{25}\,t - \frac{1}{10}\,t^2 + \frac{1}{5950}\,\mathbf{e}^{t}\,(567\cos(3\,t) + 231\sin(3\,t) - 850), \right.$$

$$z(t) = \frac{467}{250} - \frac{10904}{14875}\,\mathbf{e}^{\left(-\frac{5}{2}t\right)} - \frac{31}{50}\,t + \frac{3}{20}\,t^2 - \frac{1}{5950}\,\mathbf{e}^{t}\,(378\cos(3\,t) + 154\sin(3\,t) + 425),$$

$$\left. x(t) = \frac{148}{125} - \frac{5452}{14875}\,\mathbf{e}^{\left(-\frac{5}{2}t\right)} - \frac{53}{50}\,t + \frac{9}{20}\,t^2 + \frac{1}{5950}\,(-189\cos(3\,t) - 77\sin(3\,t) + 1275)\,\mathbf{e}^{t} \right\}$$

```
>
```